
 VB API Enhancement for Fasteners

DELMIA BWP Fastener model is exposed through IDL file for usage in Scripting. This enhancement is aimed at enriching 
the APIs to provide additional capability.
BWP already provides a few APIs to manage now, additional APIs are being provided to handle BWP fasteners.

IDL file: Interface which define APIs which can be used for a given Object for usage in Scripting language

Fastener: Fastener is a mechanical device joining two or more parts to form an assembly

Point Fastener: It is a class of fasteners which are of point type. Spot Weld, Stud Weld are examples.

Curve Fastener: It is a class of fasteners which are of line type. Glue Bead, Arc Weld are examples.  

 DELMIA BWP workbench provides IDL files for fastener objects so that users can write script utilities to automate 
repetitive tasks. 
Following is a brief description of the IDL interfaces: 

DELMIAFastener - Represents a fastener object. It has 2 sub types DELMIAPointFastener and DELMIACurveFastener.
DELMIAPointFastener - Represents a point fastener object.
DELMIACurveFastener - Represents a curve fastener object.
DELMIAFastenerGroup - Represents a fastener group object.
DELMIAFasteners - Represents a collection of fastener objects.
DELMIAFastenerItemServices - It is a utility class to assign and unassign fasteners to and from processes and resources.
DELMIAFastenerWorkBench - It is a utility class to extract all fasteners and fastenergroups from existing workbench.

Following are example use cases:

Getting fasteners and fastener groups from a workbench
Getting Fasteners from a Fastener Group 
Getting Fasteners from Products
Getting Assigned Processes from a Fastener
Getting Joining Parts from Fastener and Fastener Group
Getting Part Load Status from DPE for Fasteners
Getting DPE Defined User Attributes from a Fastener
Making Fastener ID Visible in 3D Viewer
Creating a Robot Task for Point Fasteners
Accessing Properties of Fastener Group
Accessing Properties of Fasteners
Modifying Manufacturing Positions of Fasteners
Assigning, Removing and Getting Fasteners from Processes and Resources

Getting Fasteners and Fastener groups from a workbench

Using Fastener Workbench as a technological object, we can get all the Point or Curve Fasteners or fastener 
groups available in the process document. We pass an array to the function in which all the fasteners get 
filled up. The size of array is fixed by getting the number of available fasteners using the function 
GetNumberOfPointFasteners. Here is the example script to get all point fasteners from Workbench:

'---------------------------------------------

Language="VBSCRIPT"

Sub CATMain()   

 Dim Doc As ProductDocument

 Set Doc = CATIA.ActiveDocument

 Set ItemSelection = CATIA.ActiveDocument.Selection

 

 '-- Select the Fastener

 Dim InputObjectType(0)

 InputObjectType(0) = "Product"

 Status = ItemSelection.SelectElement2(InputObjectType, "Select the Product", False)

 Set MyProduct = ItemSelection.FindObject("CATIAProduct")

 Msgbox "Selected Product = " &MyProduct.Name

 '--Getting the Workbench

 Dim wb As DELMIAFastenerWorkBench

 Set wb = MyProduct.GetTechnologicalObject("FastenerWorkBench")

 '-- Number of fasteners   

 Dim MyNum 

 MyNum = wb.GetNumberOfPointFasteners

 MsgBox MyNum

 '--Array Declaration

 Dim MyFastenerList() As AnyObject



 ReDim MyFastenerList(MyNum-1)

 '—Getting all the point fasteners

 wb.PointFasteners MyFastenerList

   For II = 0 to MyNum-1

     MsgBox MyFastenerList(II).Name

  Next

 End Sub

'---------------------------------------------

Getting Fasteners from a Fastener Group

Following is the example script to get fasteners from the selected fastener group. Recursion is used to 
get all the fasteners (from children also) of the selected group.

'---------------------------------------------

'Language="VBSCRIPT"

Dim fastnerCol() As Fastener

Dim TotalCountFastener

Dim GlobalIndex

 

Sub CATMain()

Dim objProcessDocument As Document

Set objProcessDocument = DELMIA.ActiveDocument

 

Dim objSelection As Selection

Set objSelection = objProcessDocument.Selection

 

Dim InputObjectType(0)



InputObjectType(0) = "FastenerGroup"

Status = objSelection.SelectElement2(InputObjectType, "Select the Fastener Group", 
False)

 

'---------------------------------------------

' You got the fastener group, now recursion for all fasteners

'---------------------------------------------

 

Dim MyRootFastenerGroup

Set MyRootFastenerGroup = objSelection.FindObject("DELMIAFastenerGroup")

 

CountAllFastener(MyRootFastenerGroup)

MsgBox "Number of Fasteners = " &TotalCountFastener

ReDim fastnerCol(TotalCountFastener) 

GlobalIndex = 0

CollectFasteners MyRootFastenerGroup

End Sub

 

'---------------------------------------------

Function CollectFasteners (MyFastenerGroup As FastenerGroup) 

   NumberOfFasteners = MyFastenerGroup.GetCount("DELMIAFastener")    

   

   if NumberOfFasteners > 0 Then

      Dim NFasteners ()

      ReDim NFasteners (NumberOfFasteners)



      MyFastenerGroup.GetItems "DELMIAPointFastener", NFasteners

 

      For j = 1 To NumberOfFasteners

         Dim FastenerItem As PointFastener

         Set FastenerItem = NFasteners (j-1)

         LocalIndex = GlobalIndex+j-1

         MsgBox "Fastener " &LocalIndex+1& " = " &FastenerItem.Name    

         Set fastnerCol(LocalIndex) = FastenerItem

      Next

      GlobalIndex = GlobalIndex + NumberOfFasteners

   End If

   

   NumberOfGroups = MyFastenerGroup.GetCount("DELMIAFastenerGroup")       

   Dim NGroups ()

   ReDim NGroups (NumberOfGroups)

   MyFastenerGroup.GetItems "DELMIAFastenerGroup", NGroups

   For i = 1 To NumberOfGroups

      Set Group = NGroups(i-1)

      CollectFasteners Group

   Next

End Function

 

'---------------------------------------------

Function CountAllFastener (MyFastenerGroup As FastenerGroup)

   NumberOfFasteners = MyFastenerGroup.GetCount("DELMIAFastener")    



   if NumberOfFasteners > 0 Then

      ThisSize = MyFastenerGroup.GetCount("DELMIAFastener")

      TotalCountFastener = TotalCountFastener + ThisSize

   End If

   

   NumberOfGroups = MyFastenerGroup.GetCount("DELMIAFastenerGroup")       

   Dim NGroups ()

   ReDim NGroups (NumberOfGroups)

         MyFastenerGroup.GetItems "DELMIAFastenerGroup", NGroups

   For i = 1 To NumberOfGroups

      Set Group = NGroups(i-1)

      CountAllFastener (Group)

   Next

End Function

'---------------------------------------------

Getting Fasteners from Products

Using FastenerItemServices we get all the fasteners on product(s). Also, there is an option to decide 
whether the fastener joining the products will be returned or all the fasteners on all the product(s) will 
be returned. The following script runs for pre-selected product(s).

'---------------------------------------------

Language="VBSCRIPT"  

Sub  CATMain()  

  

'Definition  of  a  list  of  types  for  objects  

Dim  listOfTypes(0)  



listOfTypes(0)="Product"  

   

     Dim  myDocument  

     Set  myDocument  =  CATIA.ActiveDocument  

   

     Dim  mySelection  As  Selection  

     Set  mySelection  =  myDocument.Selection    

 

     Dim  str  As  String  

     str=mySelection.SelectElement2(listOfTypes,"Select  a  Product",true)  

   

     Dim  number  

     number  =  mySelection.Count  

     msgbox  "Number  of  Products  selected:  "  &number    

 

Dim MyProductList() As AnyObject

ReDim MyProductList(number-1)

Dim index As Int

 

'Now,  we  retrieve  each  object  from  the  selection  

Dim  selectedElement  As  SelectedElement  

for  i=0  to  number-1

index = i+1

         Set SelectedItem = mySelection.Item(index).Value                     

         Set MyProductList(i) = SelectedItem



         MsgBox  SelectedItem.Name

next          

Set myObject = CATIA.GetItem("DNBFastenerItemServices")

MsgBox IsArray(MyProductList)

Dim MyNum 

MyNum = myObject.GetNumberOfFasteners (MyProductList, false)

MsgBox MyNum

Dim MyFastenerList() As AnyObject

ReDim MyFastenerList(MyNum-1)

myObject.GetFasteners MyProductList,false,MyFastenerList

for  j=0  to MyNum-1

    MsgBox "Fastener " &j+1& " = " &MyFastenerList(j).Name

next

End  Sub

'-------------------------------------------

Getting Assigned Processes from a Fastener

Using FastenerItemServices we can get processes to which fastener is assigned. User can get number 
of assigned processes using NumberOfAssignedProcesses. Each assigned process can be obtained 
using index as argument. Here is the script to get the process at index 1 (Index starts from 1).

'-------------------------------------------

Language="VBSCRIPT"
 
Sub CATMain()
       Dim Doc As Document
       Set Doc = CATIA.ActiveDocument
 
Set ItemSelection = CATIA.ActiveDocument.Selection
 
       '-- Select the Fastener
       Dim InputObjectType(0)
       InputObjectType(0) = "PointFastener"
       Status = ItemSelection.SelectElement2(InputObjectType, "Select the Fastener", False)
       Set MyFastener = ItemSelection.FindObject("DELMIAPointFastener")
       Msgbox "Selected Fastener = " &MyFastener.Name
       



Dim MyActivity As Activity
       Set MyActivity = MyFastener.GetProcess(1)
       MsgBox "Process = " &MyActivity.Name

 

End  Sub

'-------------------------------------------

Getting Joining Parts from Fastener and Fastener Group

Using FastenerItemServices we can get parts to which fastener is associated. User can get number of 
parts using NumberOfJoiningParts. Each part can be obtained using index as argument. Here is the 
script to get the part from fastener and fastener group at index 1 (Index starts from 1).

Getting part from a fastener at index 1:

'-------------------------------------------

Language="VBSCRIPT"
 
Sub CATMain()
       Dim Doc As Document
       Set Doc = CATIA.ActiveDocument
 
Set ItemSelection = CATIA.ActiveDocument.Selection
 
       '-- Select the Fastener
       Dim InputObjectType(0)
       InputObjectType(0) = "PointFastener"
       Status = ItemSelection.SelectElement2(InputObjectType, "Select the Fastener", False)
       Set MyFastener = ItemSelection.FindObject("DELMIAPointFastener")
       Msgbox "Selected Fastener = " &MyFastener.Name
       
Dim MyProduct As Product
Set MyProduct = MyFastener.GetPart(1)
       MsgBox "Part at index 1 = " &MyProduct.Name

End  Sub

'-------------------------------------------

Getting part from a fastener group at index 1:

'-------------------------------------------

Language="VBSCRIPT"
 
Sub CATMain()
Dim Doc As Document
Set Doc = CATIA.ActiveDocument
 
Set ItemSelection = CATIA.ActiveDocument.Selection
 
'-- Select the Activity
 Dim InputObjectType(0)



InputObjectType(0) = "FastenerGroup"
Status = ItemSelection.SelectElement2(InputObjectType, "Select the Fastener Group", 
False)
Set MyFastenerGroup = ItemSelection.FindObject("DELMIAFastenerGroup")
Msgbox "Selected Group = " & MyFastenerGroup.Name
    

 Dim MyProduct As Product

 Set MyProduct = MyFastenerGroup.GetPart(1)

 Msgbox "Part at Index 1 = " &MyProduct.Name
 
End Sub

'-------------------------------------------------------------------

Getting Part Load Status from DPE for Fasteners

All the joining parts of a fastener may not be loaded to V5 document from DELMIA Process Engineer. 
AllPartsLoaded API returns this load status of parts for fastener. 

'-------------------------------------------------------------------

Language="VBSCRIPT"
 
Sub CATMain()
    Dim Doc As Document
 Set Doc = CATIA.ActiveDocument
 
Set ItemSelection = CATIA.ActiveDocument.Selection
 
    '-- Select the Fastener
   Dim InputObjectType(0)
    InputObjectType(0) = "PointFastener"
    Status = ItemSelection.SelectElement2(InputObjectType, "Select the Fastener", 
False)
    Set MyFastener = ItemSelection.FindObject("DELMIAFastener")

    Msgbox "Selected Fastener = " &MyFastener.Name

  Dim AllPartsLoaded
  AllPartsLoaded = MyFastener.AllPartsLoaded 

      
  If (AllPartsLoaded = 0) Then
        MsgBox "ALL PARTS LOADED STATUS = FALSE"
    Else
       MsgBox " ALL PARTS LOADED STATUS = TRUE"
  End If
 
End Sub

'-------------------------------------------------------------------



Getting DPE Defined User Attributes from a Fastener

Following script is for the case where user has created in E5 a new user defined attribute, 
for spotweld planttype ,of string type, named “Fastener Eng Type” and its value is set as 
geospot. Then project is loaded in V5 and then the script is executed.

'-------------------------------------------------------------------

Language="VBSCRIPT"

 

Sub CATMain()

    Dim Doc As Document

 Set Doc = CATIA.ActiveDocument

 

Set ItemSelection = CATIA.ActiveDocument.Selection

 

    '-- Select the Fastener

   Dim InputObjectType(0)

    InputObjectType(0) = "PointFastener"

    Status = ItemSelection.SelectElement2(InputObjectType, "Select the Fastener", False)

    Set MyFastener = ItemSelection.FindObject("DELMIAPointFastener")

    Msgbox "Selected Fastener = " &MyFastener.Name

    

    Dim StrAttrVal As String

    StrAttrVal = MyFastener.GetStringUserAttribute ("Fastener Eng Type")

 

    MsgBox "Fastener Eng Type = " &StrAttrVal

 End Sub

 '-------------------------------------------------------------------



Making Fastener ID Visible in 3D Viewer

Using SetTextIdVisibilty function the text ID of the fastener is shown in the 3D window. If input to the 
function is true/1 then ID is visible and if it is false or 0 then it hides the ID.

Here is the script:

'---------------------------------------------

Language="VBSCRIPT"
 
Sub CATMain()
       Dim Doc As Document
       Set Doc = CATIA.ActiveDocument
 
Set ItemSelection = CATIA.ActiveDocument.Selection
 
       '-- Select the Fastener
       Dim InputObjectType(0)
       InputObjectType(0) = "PointFastener"
       Status = ItemSelection.SelectElement2(InputObjectType, "Select the Fastener", False)
       Set MyFastener = ItemSelection.FindObject("DELMIAPointFastener")

    Msgbox "Selected Fastener = " &MyFastener.Name

 

  '—Making Fastener ID visible

  MyFastener.SetTextIDVisibility(1) 

    End Sub

'---------------------------------------------

 


